
Establishing that a system-of-systems (SoS) architecture respects global SoSlevel properties is complex. Recording explicit technical interfaces at the boundaries of constituent systems would facilitate this, but support for the description of
such interfaces is limited in current widely-used architectural notations. This paper identifies research challenges that arise from using the combination of SysML
and the formal notation VDM to describe the interface specifications recorded at
the boundaries of the constituent systems. The approach is illustrated with a case
study based on an emergency services SoS.

Introduction

The characteristics of systems-of-systems (SoSs) identified by Maier [14] make SoS
engineering a significant challenge. At the same time, the complexity of SoSs and
the reliance placed on them is increasing. Methods and tools for SoS engineering are
therefore required in order to permit the SoS engineer to validate global SoS properties
during design and evolution. An important area of potential benefit is the description
and analysis of SoS architectures.
Although SoS engineering offers unique challenges, we believe it to be beneficial
to adapt current methods and tools as starting points [6] rather than supplanting current best practice. In our view, a promising method for SoS architectural definition
lies in interface specification, because it allows the internal definition of constituent
systems to change, as long as it continues to respect the contractual interface specification. In developing techniques for modelling SoS architectures we therefore start
from the established notation SysML [18]. While designed for systems engineering,
several features of SysML mean that it can be used for description of SoS architectures. However, like many architectural description languages, support for the formal
specification of the interfaces between the constituent systems is limited, and this in
turn limits the extent to which automated or partially automated tools can be used to
analyse semantically significant properties of models.
A possible approach to improving the level of formality is to use an existing formal
specification language in combination with SysML, allowing tools and methods developed for the formalism to be used alongside those of SysML. Formal specification
languages have a mathematically well-founded and precisely defined semantics. They
have associated techniques that allow desirable properties of a system to be specified
and demonstrated to a high degree of rigour. A suitable language is VDM [9], a formal
1

model-based specification language. The analysis techniques for VDM include static
analysis by syntax-checking, type-checking and proof, and dynamic analysis through
testing and simulation. These analyses can increase the confidence that a SoS engineer
has on the correctness of global SoS properties.
This paper identifies research challenges that arise from using SysML in combination with VDM to describe the interfaces of the constituent systems of SoSs. We
illustrate these using a case study from the emergency response domain. We begin by
summarising the current state of the art in interface contract specification in Section 2,
then introduce a case study based on an emergency response scenario in Section 3 with
the SoS architecture defined in SysML, identifying some of the interfaces in the study
and the specification of interfaces in VDM. Section 4 considers the design and analysis
of these interfaces using VDM. Section 5 concludes.

2
2.1

Interface Specification for Architectural Modelling
State of the Art in Interface Specification in Architectural Modelling

The state of the art in interface specification in architectural description notations is
poor [21]. The most widely used notations (UML [17] and SysML [18]) allow basic
signatures to be defined and pre- and postconditions to be specified textually, but these
are rarely used. In AADL [8] models are defined in terms of component types and
implementations which include subprograms (similar to operations) with basic signatures, though pre- and postconditions are not available. Formal architectural notations
such as Darwin [13] and Wright [2] allow software components to have ports defined,
and (in the case of Wright) have their message exchange protocols defined. However,
these notations do not include the ability to specify other details such as operation
pre/postconditions, and they do not contain architectural abstractions suitable for the
definition of SoSs.

2.2

Interface Contracts and Design by Contract

Interface contracts are descriptions of the constituent systems of a SoS described contractually in terms of their expectations and the obligations placed on their behaviour.
They have much in common with the idea of Design by Contract, a software engineering technique introduced by Meyer [15] in which contracts make explicit the relationships between systems in terms of preconditions and postconditions on operations and
invariants on states. In Meyer’s approach, contracts mainly specify functionality. The
interaction between operations can be described using notations such as UML sequence
diagrams or in process algebraic notation such as CSP [11] or CCS [16].
The use of contracts in service selection and subscription is an active research field
in service-oriented computing, in particular the use of contracts for the specification
of non-functional properties. Beugnard et al. [3] expand the notion of a contract to architectures in which components provide services. A four-level structure for contracts
is proposed, adding scheduling of component interaction and message passing as well
as non-functional aspects of operations. Contracts are subscribed to prior to service
invocation, after a period of negotiation.

2.3

Potential Benefits of Interface Contracts

The incorporation of interface contracts in architectural specifications may provide two
main benefits:
• Interface contracts defined for the constituent systems of a SoS architectural design permit the analysis of SoS-level properties. These analyses give SoS designers the ability to experiment with consequences of different architectural designs.
• SoS designers can define the expected interfaces of the constituent systems, and
these definitions may be provided to the system developers. This provides greater
confidence to SoS designers that constituent systems will adhere to the expected
properties on interfaces.

2.4

An Approach to Interface Contract Specification

In this paper we consider interface specification defined contractually for SoS architectural modelling. Based on the existing state of the art in architectural modelling,
identified in Section 2.1, current notations are limited for specifying interface contracts. As highlighted in Section 1, however, we believe it to be beneficial to adapt
existing notations from systems engineering practice as starting points. As such, we
propose the use of the SysML, widely-used in industry, to define SoS architectures,
along with the use of VDM to formally define interface contracts identified in SysML
models. The contribution of the paper is to identify the research challenges that arise
from using the combination of SysML with VDM to describe the interfaces of existing
constituent systems of SoSs.

3

LESLP Case Study

In order to explore the consequences of formal interface specification in a contractual
style, we use a case study based on the system formed by emergency services (fire,
police, ambulance etc.) in response to a major incident. We refer to this system as the
Major Incident Response (MIR). The MIR is a SoS in Maier’s terms [14] and may be
considered an acknowledged SoS [7]. The emergency services are operationally and
managerially independent. Each service is itself geographically distributed and may
evolve, for example as personnel come on and off duty during the course of a longrunning incident. Emergent behaviour is also present – the comprehensive approach
to management of the incident relies on voluntary and collaborative interaction. We
give more details on the case study in Section 3.1, and an architectural description in
Section 3.2. Section 3.3 supplements this description with a formal definition of the
interfaces.

3.1

Informal Description of the Case Study

We base the study on the procedures for the coordination of the MIR in London, as
outlined in the Major Incident Procedure Manual [19] published by the London Emergency Services Liaison Panel (LESLP). This documents the process for identifying a
major incident, the initial information to be passed to the appropriate services and the
roles and responsibilities of the service members at the scene.

The response to all major incidents follows a broadly similar structure. The members from each service attending the scene form Bronze command. For more severe
incidents, a Silver command will be formed containing representatives of all the involved services. For long-running incidents, a Gold command may be formed at a
geographically distant point. The Bronze, Silver, Gold hierarchy corresponds to the
operational, tactical and strategic levels of command.
We pay particular attention to the rules outlined in [19] for communication of casualty information with the media, and the requirements these place on the interfaces
between the emergency services. In the early stages of a major incident, confusion
can arise if the media aggregate casualty figures from various sources. This can lead
to “double-counting” and overestimation of the severity of incident. To avoid this, all
casualty details must be given to Gold command, which is then responsible for coming
to a more reliable estimate and communicating this to the media.
In [5] the case study was explored using the Event-B [1] formalism, but the model
developed there does not provide an accessible representation of the SoS architecture,
and does not consider the interface specification between the constituent systems. In
this paper we focus on these aspects.

3.2

Architectural Description of LESLP

SysML [18] is a profile for UML 2.0, developed for system engineering, but also supporting the modelling of SoS architecture. It enjoys wide industrial support and a sound
tool base. SysML provides several diagram types, with a “precise natural language”
semantics, to support the description of the SoS structure, behaviour and requirements.
The MIR structure of the case study is given in Figure 1 as a SysML Block Definition Diagram (BDD). The MIR contains up to three emergency services and these are
the constituent systems of the MIR SoS. The emergency services (ES) may be a police
force, a fire brigade or ambulance service. All ES contain one or more person and each
person has a role (Bronze, Silver or Gold).
We consider one requirement of the MIR SoS, that only accurate casualty information should be released to the media, and show in detail how the constituent systems
communicate1 . These communications between instances of the constituent systems
with a given role are described in an Internal Block Diagram (IBD) given in Figure 2.
The IBD details the provided and required interfaces of the systems of the MIR SoS
and depends upon the roles undertaken by the constituents. For example, in the IBD
of Figure 2, an interface info to silver exists between Officers with a Bronze role and
their respective emergency service Officers with Silver role.
The SysML interface definitions in Figure 3 relate to points of interaction for those
operations made public by the relevant constituent systems. The operations are defined
in terms of operation signatures detailing data input to and output from an operation
call. For example, the info to silver interface, relating to the transferring of casualty
information from Bronze officers to Silver Ambulance officers, consists of the operation verifyCasualtyDetails(CasualtyDetails) which requires some unverified casualty
information of the CasualtyDetails data type (defined elsewhere in the model).
The SysML specification allows pre- and postconditions to be specified for operations, however this is optional and no analyses are available to ensure their correctness.
SysML also omits the Protocol State Machine of UML 2.0 which dictates the response
of an interface to specified sequences of events, constraining the order of operations. It
1 We

consider only the communications and interactions that are necessary to meet this requirement.

bdd [MIR Structure] Major Incident Response Structure
«block»
Major Incident Response

1..*
1..3
«block»
Emergency Service

«block»
Police Service

1

«block»
Person

1..*

«block»
Police Officer

«block»
Ambulance Service

«block»
Fire Service

«block»
Ambulance Officer

«block»
Fire Officer

Figure 1: Block definition diagram depicting Major Incident Response structure.
ibd [Major Incident Response] Casualty Info Clearance
: Major Incident Response
info_to_clear
press_conf

gold : Police Officer

gold : Ambulance Officer
info_to_gold

info_to_verify
info_to_verify

: Media

silver : Police Officer

silver : Ambulance Officer

info_to_silver

info_to_silver

bronze : Police Officer

bronze : Ambulance Officer

silver : Fire Officer

info_to_silver

bronze : Fire Officer

!

Figure 2: Internal Block Diagram showing relationships of response constituents when
releasing casualty figures.
is our opinion that this construct would add additional rigour to an interface specification and increase the range of analyses available.

bdd [Interfaces] Casualty Figures
«interface»
info_to_silver

«interface»
info_to_verify

receiveCasualtyDetails (in CasualtyDetails)

verifyCasualtyDetails (in CasualtyDetails)

«interface»
info_to_gold

«interface»
info_to_clear

receiveCasualtyDetails (in CasualtyDetails)

clearCasualtyDeatils (in CasualtyDetails)

«interface»
press_conf
pressRelease () : ClearedDetails

Figure 3: Block Definition Diagram showing interface definitions when releasing casualty figures.

3.3

Formal Definition of LESLP Interface Contracts

Given a SysML architectural specification, we strengthen the interface specifications of
the constituent systems as described above. Using a formal model-based specification
language allows increased confidence using a range of analysis techniques. In this
paper, VDM is used to give formal definitions of the architectural interfaces. VDM is
a model-oriented notation supporting descriptions of data and functionality. VDM has
industry-strength2 and open-source3 tool support.
The SysML interfaces in Figure 3 may be modelled as VDM classes. VDM supports inheritance, so we define the operations of interfaces with pre- and postconditions,
and provide no operation body at the interface. The operation definition, therefore, does
not state what is done, the implementation of the algorithm must be provided by the
constituent systems implementing the interface. This is denoted using the is subclass
responsibility key phrase in the operation body. The system designer must give a specification for each constituent system which provides the interface, and ensure that the
implementation of the constituent system contains the interface operations, and that
they are consistent with the interface specification.
Figures 4a) and 4b) define interface definitions of the info to clear and press conf
interfaces initially given in Figure 3. The operations keyword in each class denotes the
available operations of the interfaces with signatures corresponding to those defined in
Figure 3. Each operation is strengthened using pre- and postconditions. The clearCasualtyDetails operation of the info to clear interface, shown in Figure 4a), requires a
parameter cd, of type CasualtyDetails. The operation does not return any result. As
mentioned above, no operation body is given. The operation precondition, denoted by
2 VDMTools:
3 Overture:

http://www.vdmtools.jp/en
http://www.overturetool.org

the keyword pre, states that there must be 0 or greater reported casualties (cd.number
>= 0), they must have been verified (cd.verified), and are not already cleared for release (not cd.cleared). No postcondition is provided.
The press conf interface in Figure 4b) contains one operation, pressRelease. The
operation has no parameters and returns a result of type ClearedDetails. This interface
operation also has no body. The operation has no precondition, but has a postcondition,
which is given after the post keyword. We may refer to the return variable using the
RESULT keyword. The postcondition requires that the result has been verified and
cleared (RESULT.verified and RESULT.cleared).

class info_to_clear
operations
public clearCasualtyDetails: CasualtyDetails ==> ()
clearCasualtyDetails(cd) ==
is subclass responsibility
pre cd.number >= 0 and cd.verified and
not cd.cleared
end info_to_clear





(a) info to clear interface.


class press_conf

operations
public pressRelease: () ==> ClearedDetails
pressRelease () ==
is subclass responsibility
post RESULT.verified and RESULT.cleared;
end press_conf





(b) press conf interface.

Figure 4: VDM representations of interface definitions.
Given these interface specifications, VDM classes are defined which implement
the various interface definitions to correspond with the IBD in Figure 2. These classes
are implemented using the is subclass of key phrase followed by the interface class
name. The classes (for example Gold Police) also implement the Police abstract class
detailing emergency service-specific variables.
The Gold Police system is defined as a VDM class below in Figures 5, 6 and 7. Figure 5 shows the Gold Police class which implements the info to clear and press conf
interfaces, defined on the first two lines. The class has a single instance variable,
cas cleared, a set of type CasualtyDetails, initially set to the empty set {}.
In Figure 6, the Gold Police class implements the clearCasualtyDetails operation,
as defined in the info to clear interface. The operation strengthens the precondition to
ensure that the police officer has the rank Gold (role = <Gold>) and the remainder of
the precondition is unchanged. The operation postcondition ensures that the number
of CasualtyDetails items in the cas cleared instance variable does not decrease (card
cas cleared >= card cas cleared˜)4 . The postcondition also states that, for all CasualtyDetails items in the cas cleared set which have the same location as the parameter
4 In VDM postconditions, a variable name succeeded by the ˜ symbol refers to the initial value of that
variable.

class Gold_Police is subclass of Police,
info_to_clear, press_conf
instance variables
private cas_cleared : set of CasualtyDetails := {};
...
end Gold_Police





Figure 5: VDM class representing Gold Police constituent system: preamble and instance variables.
cd, those details should be verified and cleared (forall c in set cas cleared & c.loc =
cd.loc => c.verified and c.cleared).

class Gold_Police
...
operations
public clearCasualtyDetails : CasualtyDetails ==> ()
clearCasualtyDetails(cd) ==
(
if not exists c in set cas_cleared & c.loc = cd.loc
then cas_cleared := cas_cleared union
{mk_CasualtyDetails(cd.number, true,
true, cd.type, cd.loc)}
)
pre role = <Gold> and cd.number >= 0 and
cd.verified and not cd.cleared
post card cas_cleared >= card cas_cleared˜ and
forall c in set cas_cleared & c.loc = cd.loc =>
c.verified and c.cleared;
...
end Gold_Police



Figure 6: VDM class representing Gold Police constituent system: clearCasualtyDetails operation.
The operation body of the clearCasualtyDetails operation, given after the operation
signature, has an if statement as the main structure. If there does not exist any casualty
details in the cas cleared set with the same location as the parameter cd (if not exists
c in set cas cleared & c.loc = cd.loc then), then a new CasualtyDetails item is added
to the cas cleared set with the verified and cleared fields set to true, all other fields
are given as those supplied by the parameter, cd (cas cleared := cas cleared union
{mk CasualtyDetails(cd.number, true, true, cd.type, cd.loc)}).
The Gold Police class also implements the press conf interface and so, in Figure 7,
the pressRelease operation is given. The pressRelease operation in Figure 7 has a
strengthened precondition, as with the clearCasualtyDetails operation, to ensure that
police officers carrying out the operation have the rank <Gold>. The postcondition is
also strengthened to ensure that the number of casualty details released is less than, or
equal to, the number of cleared details the Gold Police know about. The operation uses
a private function, totalCleared, to calculate the total cleared figure (RESULT.number
8



<= totalCleared(cas cleared)). This is an important property of the casualty clearance scenario – that the Major Incident Response command do not release casualty
figures exceeding the number of casualties that have been verified and cleared for release. The Gold Police are able to use their discretion in releasing a lower figure than
is known. The pressRelease operation body passes the cas cleared instance variable
to a private clearForPress function, the result of which is returned (return clearForPress(cas cleared)). The private clearForPress and totalCleared functions are given in
Figure 7. The clearForPress operation body is undefined using the is not yet specified
key phrase – a policy decision for the Gold Police, not given here. Finally, the totalCleared function is a simple recursive function to count the number of casualties given
a CasualtyDetails set.

class Gold_Police
...
operations
public pressRelease : () ==> ClearedDetails
pressRelease () ==
(
return clearForPress(cas_cleared);
)
pre role = <Gold>
post RESULT.cleared and RESULT.verified and
RESULT.number <= totalCleared(cas_cleared);
functions
private clearForPress : set of CasualtyDetails
==> ClearedDetails
clearForPress(cds) ==
is not yet specified;
private totalCleared : set of CasualtyDetails -> nat
totalCleared(cds) ==
cases cds:
{} -> 0,
others -> let cd in set cds in
cd.number + totalCleared(cds\{cd})
end;
end Gold_Police





Figure 7: VDM class representing Gold Police constituent system: pressRelease operation and auxiliary private functions.
The remainder of the VDM model of the Major Incident Response is given in Appendix A

4

Analysis of Interface Contracts

The purpose of a formal analysis is to confirm or refute specified properties that are
required of the model under consideration. The analysis techniques available for VDM
include static and dynamic techniques, with varying degrees of machine support. VDM’s
particular history of use in industry settings requiring extensive test-based validation

of models and model-based testing of implementations mean that its tool sets have
highly developed interpreters allowing rapid testing of models on high volumes of test
cases [12]. Simulation is closely linked to testing. Rather than executing a single welldefined test, the model execution is driven by a scenario containing multiple decision
points that may resolved by a user interacting with the model. This allows those not
experienced in the notation or involved in the development to gain familiarity with the
formal model, and provides a valuable way of exposing application domain experts to
the model at an early stage.
VDM has a well-defined formal semantics, and therefore VDM models are amenable
to logical proof [4]. Proof obligations arise naturally within a model, for example the
obligation to prove that the specification of each operation is satisfiable, i.e., for all
valid pre-states and inputs there is always a state of the model that satisfies the postcondition of the operation. Overture generates proof obligations automatically, and
manages their manual “sign off” by the user, but currently little help is given to the
user in automatically discharging them.
Testing, simulation and proof both have a contribution to make in increasing our
assurance of the design of the MIR SoS. For example, both proof and unit testing of the
operations would be valuable to ensure that explicit definitions for operations meet the
implicit (pre- and postcondition) specifications. Proof is an expensive technique, and
best applied only to a small number of key properties, but it provides a higher degree
of confidence in a system. The MIR SoS was designed to ensure that only verified
casualty figures were released to the media. Some properties relating to this purpose are
suggested in [20], for example that only Police Gold is authorised to release casualty
figures, or that Unverified casualty figures must never be released to the general public.
Demonstrating properties such as these would be an appropriate application of proof
technology.

5

Conclusion

From the work documented in this paper, we conclude certain requirements on the notation used for the specification of interfaces between constituent systems. Significant
requirements include the necessity of using a formal notation which includes architectural abstractions and the ability to describe the accepted orderings of events at the
interface. We also observe the necessity to provide strong ties to an accepted industrial
strength architectural description language and the ability to deal with different levels
of abstraction.
Whilst SysML enables SoS engineers to model complex SoS architectures, the facilities for interface definition are less satisfactory and little analysis is possible. Using
a formal notation enables these analyses. However, existing formal notations with architectural abstractions are poor. We propose the use of formal specification notations,
such as VDM, to define interfaces and concrete system specifications and to reason
about their properties. In the paper, we demonstrate the use of VDM by defining interfaces corresponding to the SysML architectural model, and further specify systems
implementing the interfaces. Whilst not performed in this paper, VDM allows strong
static analysis support and dynamic support in the form of simulation, testing and proof
obligation generation, in particular allowing the specification of interfaces at different
levels of abstraction.
Current VDM tools lack model checking and proof support. Further, although it
supports data-based specification of functionality, VDM does not contain abstractions
10

to support description of event orderings at interfaces. Existing notations which could
fill these gaps in analyses include the family of process algebras, e.g. CSP. This would
allow the definition of protocols on interfaces and provide dynamic analysis through
model checking, which could be used to avoid protocol mismatch [10]. An optimal
approach however, would be to use a notation that provides data-based modelling (as
in VDM) and event ordering (such as CSP) and which also contains the abstractions
necessary to model SoS architectures and state and reason over global SoS properties.
The development of such a notation is the goal of our current work in the COMPASS project5 . This project aims to improve the state of the art in SoS engineering by
provision of modelling tools and analysis techniques based on an underlying modelling
language (the COMPASS Modelling Langage, CML). CML provides VDM-style data
modelling and CSP-style event ordering as outlined above, for representing SoS architectures and interface contracts. This language will have a formal semantic definition
to support description of behaviour and composition of subsystem properties, based
on Hoare and He’s Unifying Theories of Programming (UTP) [?] and will integrate
with SysML to support modelling using either textual CML, graphical SysML or a
combination of the two.

A

VDM Models of LESLP Case Study

A.1

Interface Definitions

A.1.1

Interface Datatypes


class IF_Types
-- Basic Type definitions to be used in the Major Incident Model -types
public CasualtyDetails :: number : nat
verified : bool
cleared : bool
type : token
loc : Location;
public ClearedDetails :: number : nat
verified : bool
cleared : bool
type : set of token;
public Location = token;
end IF_Types



A.1.2



info to silver


class info_to_silver
operations
-- Receive Collected Casualty Details operation
--- parameters:
cd : CasualtyDetails
-- precondition: ensures the supplied cd parameter has not been
-verified or cleared
public receiveCollectedCasualtyDetails : IF_Types‘CasualtyDetails ==> ()
receiveCollectedCasualtyDetails(cd) ==
is subclass responsibility
pre not(cd.verified) and not(cd.cleared) and cd.number >= 0;
end info_to_silver



A.1.3



info to verify


class info_to_verify
operations
-- Verify Casualty Details operation
--- parameters:
cd : CasualtyDetails
-- precondition: ensures the supplied cd parameter has not been

-verified or cleared, and the details have been
-collected
public verifyCasualtyDetails : IF_Types‘CasualtyDetails ==> ()
verifyCasualtyDetails(cd) ==
is subclass responsibility
pre not(cd.verified) and not(cd.cleared) and cd.number >= 0
end info_to_verify



A.1.4



info to gold


class info_to_gold
operations
-- Receive Casualty Details operation.
--- parameters:
cd : CasualtyDetails
-- precondition: the casualty details have been verified, not
-cleared and greater than or equal to 0
public receiveVerifiedCasualtyDetails : IF_Types‘CasualtyDetails ==> ()
receiveVerifiedCasualtyDetails(cd) ==
is subclass responsibility
pre cd.number >= 0 and cd.verified and not cd.cleared
end info_to_gold





A.1.5

info to clear


class info_to_clear
operations
-- Receive Casualty Details operation.
--- parameters:
cd : CasualtyDetails
-- precondition: the casualty details have been verified, not
-cleared and greater than or equal to 0
public clearCasualtyDetails : IF_Types‘CasualtyDetails ==> ()
clearCasualtyDetails(cd) ==
is subclass responsibility
pre cd.number >= 0 and cd.verified and not cd.cleared
end info_to_clear



A.1.6



press conf


class press_conf
operations
-- Press Release operation
--

-- returns:
res : ClearedDetails
-- postcondition: Cleared details are marked as cleared and verified
public pressRelease : () ==> IF_Types‘ClearedDetails
pressRelease () ==
is subclass responsibility
post RESULT.cleared and RESULT.verified;
end press_conf



A.1.7



order to collect info


class order_to_collect_info
operations
-- Collect Casualty Details operation
--- parameters:
loc: Location
public collectCasualtyDetails: IF_Types‘Location ==> ()
collectCasualtyDetails(loc) ==
is subclass responsibility
end order_to_collect_info





A.2

Constituent System Types

A.2.1

Police


class Police is subclass of Person
types
public PoliceId = token;
public PoliceRank = <Constable> | <Sergeant> | <Inspector> |
<Commander> | <Commissioner> | <UnAssigned>;
instance variables
protected id : PoliceId;
protected rank : PoliceRank;
operations
-- Constructor ------------------public Police : PoliceId * IF_Types‘Location ==> Police
Police(pid, l) ==
(
id := pid;
rank := <UnAssigned>;
role := <UnAssigned>;
location := l;
);
public getId : () ==> PoliceId
getId() == return id;

end Police



A.2.2



Ambulance


class Ambulance is subclass of Person
types
public AmbulanceId = token;
public AmbulanceRank = <Paramedic> | <AmbulanceManager> |
<UnAssigned>;
instance variables
protected id : AmbulanceId;
protected rank : AmbulanceRank;
operations
-- Constructor ------------------public Ambulance : AmbulanceId * IF_Types‘Location ==> Ambulance
Ambulance(aid, l) ==
(
id := aid;
rank := <UnAssigned>;
role := <UnAssigned>;
location := l;
);
public getId : () ==> AmbulanceId
getId() == return id;
end Ambulance



A.2.3

Fire


class Fire is subclass of Person
types
public FireId = token;
public FireRank = <Firefighter> | <CrewManager>| <Commissioner> |
<UnAssigned>;
instance variables
protected id : FireId;
protected rank : FireRank;
operations
-- Constructor ------------------public Fire : FireId * IF_Types‘Location ==> Fire
Fire(fid, l) ==

(
id := fid;
rank := <UnAssigned>;
role := <UnAssigned>;
location := l;
);
public getId : () ==> FireId
getId() == return id;
end Fire





A.3

Constituent System Definitions

A.3.1

Gold Police


class Gold_Police is subclass of Police, press_conf, info_to_clear
instance variables
private cas_cleared : set of IF_Types‘CasualtyDetails;
operations
-- Constructor ------------------public Gold_Police : PoliceId * PoliceRank * Role * IF_Types‘Location
==> Gold_Police
Gold_Police(pid, ra, ro, l) ==
(
id := pid;
rank := ra;
role := ro;
location := l;
cas_cleared := {}
);
-- info_to_clear interface operations --------------------------------------public clearCasualtyDetails : IF_Types‘CasualtyDetails ==> ()
clearCasualtyDetails(cd) ==
(
if forall c in set cas_cleared & c.loc <> cd.loc
then cas_cleared := cas_cleared union
{mk_IF_Types‘CasualtyDetails(cd.number, true,
true, cd.type, cd.loc)}
)
pre role = <Gold> and cd.number >= 0 and cd.verified and
not cd.cleared
post card cas_cleared >= card cas_cleared˜ and
forall c in set cas_cleared &
c.loc = cd.loc => c.verified and c.cleared;

-- press_conf interface operations -------------------------------------public pressRelease : () ==> IF_Types‘ClearedDetails
pressRelease () ==
(
return clearForPress(cas_cleared);

)
pre role = <Gold>
post RESULT.cleared and RESULT.verified and
RESULT.number <= totalCleared(cas_cleared);

-- Private operations -------------------------- policy for determining what data to release to press
private clearForPress : set of IF_Types‘CasualtyDetails ==>
IF_Types‘ClearedDetails
clearForPress(cds) ==
is not yet specified;

-- Auxilliary functions -------------------------functions
-- function to count the number of casualty details cleared by Gold
private totalCleared : set of IF_Types‘CasualtyDetails -> nat
totalCleared(cds) ==
cases cds:
{} -> 0,
others -> let cd in set cds in
cd.number + totalCleared(cds\{cd})
end;
end Gold_Police



A.3.2



Gold Ambulance


class Gold_Ambulance is subclass of Ambulance, info_to_gold
instance variables
private cas_verified : set of IF_Types‘CasualtyDetails;

operations
-- Constructor ------------------public Gold_Ambulance : AmbulanceId * AmbulanceRank * Role *
IF_Types‘Location ==> Gold_Ambulance
Gold_Ambulance(aid, ra, ro, l) ==
(
id := aid;
rank := ra;
role := ro;
location := l;
cas_verified := {}
);

-- info_to_gold interface operations --------------------------------------public receiveCasualtyDetails : IF_Types‘CasualtyDetails ==> ()
receiveCasualtyDetails(cd) ==
(

18

cas_verified := cas_verified union {cd}
)
pre role = <Gold> and cd.number >= 0 and cd.verified and
not cd.cleared
post cd in set cas_verified and cd.verified and not cd.cleared;
end Gold_Ambulance





A.3.3

Silver Ambulance


class Silver_Ambulance is subclass of Ambulance, info_to_silver,
info_to_verify
instance variables
-- collection of verified casualty details
private cas_verified : set of IF_Types‘CasualtyDetails := {};
private cas_estimates : set of IF_Types‘CasualtyDetails := {};
operations
-- Constructor ------------------public Silver_Ambulance : AmbulanceId * AmbulanceRank * Role *
IF_Types‘Location ==> Silver_Ambulance
Silver_Ambulance(aid, ra, ro, l) ==
(
id := aid;
rank := ra;
role := ro;
location := l;
cas_verified := {}
);
-- info_to_silver interface operations -----------------------------------------public receiveCollectedCasualtyDetails : IF_Types‘CasualtyDetails
==> ()
receiveCollectedCasualtyDetails(cd) ==
(
cas_estimates := cas_estimates union {cd};
verifyCasualtyDetails(cd)
)
pre not(cd.verified) and not(cd.cleared) and cd.number >= 0;
-- info_to_verify interface operations -----------------------------------------public verifyCasualtyDetails : IF_Types‘CasualtyDetails ==> ()
verifyCasualtyDetails( cd) ==
if forall c in set cas_verified & c.loc <> cd.loc
then cas_verified := cas_verified union
{mk_IF_Types‘CasualtyDetails(cd.number, true,
cd.cleared, cd.type, cd.loc)}
pre role = <Silver> and not cd.verified and cd.number >= 0 and
not cd.cleared
post exists c in set cas_verified &
c.loc = cd.loc and c.verified and not(c.cleared) and
forall cas in set (cas_verified˜\{c}) &
forall cas1 in set (cas_verified\{c}) & cas = cas1;

end Silver_Ambulance



A.3.4



Bronze Police


class Bronze_Police is subclass of Police, order_to_collect_info
instance variables
-- casualty details collected by officer
private cas_estimates : set of IF_Types‘CasualtyDetails;
operations
-- Constructor ------------------public Bronze_Police :

PoliceId * PoliceRank * Role *
IF_Types‘Location ==> Bronze_Police
Bronze_Police(aid, ra, ro, l) ==
(
id := aid;
rank := ra;
role := ro;
location := l;
cas_estimates := {};
);
-- Collect interface operations ----------------------------------public collectCasualtyFigures : IF_Types‘Location ==> ()
collectCasualtyFigures(loc) ==
cas_estimates := cas_estimates union {getCasualtyEstimates(loc)}
post exists c in set cas_estimates & c.loc = loc and
not(c.verified) and not(c.cleared) and c.number >= 0;
-- Operation to collect casualty estimates
private getCasualtyEstimates : IF_Types‘Location ==>
IF_Types‘CasualtyDetails
getCasualtyEstimates(loc) ==
(
let cas_no = collect(loc) in
return mk_IF_Types‘CasualtyDetails(cas_no, false, false,
mk_token(""), loc);
);
--operation to collect casualty numbers
private collect : IF_Types‘Location ==> nat
collect(loc) ==
is not yet specified;

end Bronze_Police



A.3.5



Bronze Fire


class Bronze_Fire is subclass of Fire, order_to_collect_info

instance variables
-- casualty details collected by officer
private cas_estimates : set of IF_Types‘CasualtyDetails;
operations
-- Constructor ------------------public Bronze_Fire : FireId * FireRank * Role *

IF_Types‘Location
==> Bronze_Fire

Bronze_Fire(aid, ra, ro, l) ==
(
id := aid;
rank := ra;
role := ro;
location := l;
cas_estimates := {};
);
-- Collect interface operations ----------------------------------public collectCasualtyFigures : IF_Types‘Location ==> ()
collectCasualtyFigures(loc) ==
cas_estimates := cas_estimates union {getCasualtyEstimates(loc)}
post exists c in set cas_estimates & c.loc = loc and
not(c.verified) and not(c.cleared) and c.number >= 0;

-- Operation to collect casualty estimates
private getCasualtyEstimates : IF_Types‘Location ==>
IF_Types‘CasualtyDetails
getCasualtyEstimates(loc) ==
(
let cas_no = collect(loc) in
return mk_IF_Types‘CasualtyDetails(cas_no, false, false,
mk_token(""), loc);
);
--operation to collect casualty numbers
private collect : IF_Types‘Location ==> nat
collect(loc) ==
is not yet specified;
end Bronze_Fire



A.3.6



Bronze Ambulance


class Bronze_Ambulance is subclass of Ambulance, order_to_collect_info
instance variables
-- casualty details collected by officer
private cas_estimates : set of IF_Types‘CasualtyDetails;
operations
-- Constructor ------------------public Bronze_Ambulance : AmbulanceId * AmbulanceRank * Role *
IF_Types‘Location ==> Bronze_Ambulance
Bronze_Ambulance(aid, ra, ro, l) ==

(
id := aid;
rank := ra;
role := ro;
location := l;
cas_estimates := {};
);
-- Collect interface operations ----------------------------------public collectCasualtyFigures : IF_Types‘Location ==> ()
collectCasualtyFigures(loc) ==
cas_estimates := cas_estimates union {getCasualtyEstimates(loc)}
post exists c in set cas_estimates & c.loc = loc and
not(c.verified) and not(c.cleared) and c.number >= 0;

-- Operation to collect casualty estimates
private getCasualtyEstimates : IF_Types‘Location ==>
IF_Types‘CasualtyDetails
getCasualtyEstimates(loc) ==
(
let cas_no = collect(loc) in
return mk_IF_Types‘CasualtyDetails(cas_no,false, false,
mk_token(""), loc);
);
--operation to collect casualty numbers
private collect : IF_Types‘Location ==> nat
collect(loc) ==
is not yet specified;
end Bronze_Ambulance

